

```

1 1 SURVO MM Fri Sep 28 23:36:50 2007 D:\SURVO\GRAF\ 1000 200 0
218 *
219 *Järjestetään aineisto ja talletetaan lajitteluavain muuttujaksi Aika:
220 *
221 *FILE SORT KUVAT2 BY Vuosi,Kausi TO KUVAT3 / KEY_SAVED=Aika
222 *FILE SHOW KUVAT3
223 *
224 *GLOT KUVAT3 Aika,Koko
225 *.....
226 *
227 *FILE AGGR KUVAT3 BY Aika TO KUVAT3A / siirrytään kuukausitason aineistoon
228 *VARIABLES:
229 *Aika FIRST Aika / Jos olisi Koko = MISSING, niin nuo havainnot
230 *Kausi FIRST Kausi / eivät tulisi mukaan aggregoituun dataan. Siksi
231 *Lkm N - / 'placeholder'-havainnoissa Koko = 0. Mutta tällöin
232 *Koko:8 SUM Koko / aggregaatin lukumäärät on syytä korjata (vähentää 1)
233 *END / ja laskea keskkoot vasta sen jälkeen (siksi SUM).
234 *
235 *.....
236 *FILE SHOW KUVAT3A
237 *VAR i:1,Lkm / Lkm=Lkm-1 i=ORDER
238 *.....
239 *VAR Koko=Koko/Lkm / SELECT=!0 jossa 0=Koko,0 / keskkoot
240 *.....
241 *Poistetaan ensimmäisen ja viimeisen havainnon Lkm-tiedot:
242 *VAR Lkm=MISSING / CASES=Aika:200101,200612
243 *(syy: kamera ostettu 2/2001, viimeiset kuvat otettu 11/2006)
244 *.....
245 *
246 *Piirretään kuva PostScript-muotoon, siitä EPS- ja edelleen PDF-muotoon:
247 *VAR väri:2=-Kausi TO KUVAT3A / värikoodien (ks. alla) piirtämistä varten
248 *
249 *PLOT KUVAT3A i,Lkm / LINE=1 POINT=0,10,Koko,400000 FRAME=6 HEADER=
250 *XSCALE=[move(70,30)],-1:?,1:2001,13:2002,25:2003,37:2004,49:2005,61:2006,73:
251 *YSCALE=[Swiss(10)],-30:?,0(100)500 XLABEL= TICKLENGTH=20
252 *YLABEL=[TimesI(13)][color(0,0,0,0.5)][move(70,-90)],&
253 * Canon_Digital_Ixus_-kameralla_otetut_kuvat
254 *
255 *LINETYPE=[Swiss(12)][move(0,0)][BLACK]
256 * PEN=[Swiss(12)][move(0,0)][BLACK]
257 *DEVICE=PS,KUVA.PS SIZE=1600,650 XDIV=1,15,0 YDIV=100,500,50
258 *
259 *Kuvataan vuodenaikojen vaihtelua käyttäen täsmennystä POINT_COLOR=väri jossa
260 *väri on edellä luotu muuttuja ja värivalinnat toteutettu seuraavin tavoin:
261 *
262 * FILL(-1)=*talvi Kullekin vuodenaajalle oma CMYK-sävy:
263 * FILL(-2)=*talvi *talvi=1.0,1.0,0.0,0.0 (sininen)
264 * FILL(-3)=*kevät *kevät=0.6,1.0,0.6,0.0 (violetti)
265 * FILL(-4)=*kevät *kesä!=1.0,0.0,1.0,0.0 (vihreä)
266 * FILL(-5)=*kevät *syksy=0.0,0.4,0.4,0.3 (ruskea)
267 * FILL(-6)=*kesä!
268 * FILL(-7)=*kesä! FILL(-10)=*syksy
269 * FILL(-8)=*kesä! FILL(-11)=*syksy
270 * FILL(-9)=*syksy FILL(-12)=*talvi
271 *
272 *DATA SELITE
273 *väri x y z
274 * -1 0 1 talvi
275 * -3 3 1 kevät
276 * -6 6 1 kesä
277 * -9 9 1 syksy
278 *
279 *PLOT SELITE,x,y / POINT=[SwissB(11)],z LINE=0 XSCALE=0:?,10:? YSCALE=0:?,1:? ...
280 *EPS JOIN K 1 2 / 1=KUVA 2=SELITE,250,400
281 */GV-SHOW K.PS / haetaan sopivat BOUNDINGBOX-arvot
282 *
283 *EPS K.PS K.EPS / BOUNDINGBOX=230,150,1860,720 (yksikkönä 0.1mm)
284 *>EPStoPDF K.EPS --outfile=CanonDigitalIxsu2001-2006.pdf
285 */OPEN CanonDigitalIxsu2001-2006.pdf
286 *
... 279: YLABEL= SIZE=600,100 XDIV=0,1,0 YDIV=0,1,0 FRAME=0 DEVICE=PS,SELITE.PS

```